# Maintainer: Levente Polyak <anthraxx[at]archlinux[dot]org>

_pyname=sphinxcontrib-log-cabinet
pkgname=python-${_pyname}
pkgver=1.0.1
pkgrel=4
pkgdesc='Organize changelog directives in Sphinx docs'
url='https://github.com/davidism/sphinxcontrib-log-cabinet'
arch=('any')
license=('BSD')
depends=('python' 'python-sphinx' 'python-setuptools')
source=("${url}/archive/${pkgver}/${pkgname}-${pkgver}.tar.gz")
sha512sums=('1f8c6a0b2c7432b42eb07246f5034b93724b90b6e2a5de3f77c3e72c6e928e150373b1abb553d601c4820fefe8b0dd72d1301fdd82366528c45ae3a158adb97d')
b2sums=('5df31c73590ebebd7c51cd0234e90f27d2034a0edd24cb4a3da875199e9357071b8adb63643c920cd185384e474fc0480b0943d4ac07b398d8ff4835571200a5')

build() {
  cd ${_pyname}-${pkgver}
  python setup.py build
}

package() {
  cd ${_pyname}-${pkgver}
  python setup.py install --root="${pkgdir}" -O1 --skip-build
  install -Dm 644 LICENSE.rst -t "${pkgdir}/usr/share/licenses/${pkgname}"
  install -Dm 644 CHANGES.rst README.rst  -t "${pkgdir}/usr/share/doc/${pkgname}"
}

# vim: ts=2 sw=2 et:
